Mathematical Modelling with Analytic Tools

This chapter is about differential equations. My university professor once said that indeed everything was about algebra in mathematics but it was hard to notice it. This statement aroused my interest in this field once and for all. However, the co-author of this book seems to have attended another university thus in this chapter we are going to show the strengths and mathematical know-how of Maple concerning differential equations.

8.1 Orthogonal Trajectories

Consider the ellipses in the plane determined by the

$$x^2 + 3\,y^2 = c$$

equation.

By changing the value of the c parameter located in the equation we can get a family of ellipses which is called a class of ellipses. Look for the equation of the class of curves that contains one free parameter in the plane of the ellipses and the arbitrary curve of which perpendicularly crosses any of the ellipses. In other words, the orthogonal trajectory of the class of ellipses with one parameter is to be determined.

Draw the ellipses and their orthogonal trajectories in the same coordinate system.

It is not easy to imagine two classes each of which contains infinite amount of curves and the elements of which perpendicularly crosses each other in pairs. This means that if we consider an arbitrary element of a class of curves then it is crossed perpendicularly by many infinite elements of the other class of curves. This is horrible but we do hope that by the end of the worksheet we will be more enthusiastic about this.

For the better understanding of orthogonal trajectories let's see an easier task. Examine the class of lines determined by the y=mx equation. The elements of the class of lines are lines which go through the origin with different slopes. It is obvious that every concentric circle with a [képlet] origin centre perpendicularly crosses any of these lines because it is known that the tangent of the circle is perpendicular to the line which connects the point of tangency with the centre of the circle. According to this, the orthogonal trajectory of the y=mx one-parameter class of lines is the one-parameter class of concentric circles that can be described by the [képlet] equation. Draw these in the same coordinate system to illustrate the two perpendicular classes of curves.

> *restart*

> $e := seq\left(plot\left(\left[\cos\left(\frac{1}{10}\,m\,\mathrm{Pi}\right)t,\ \sin\left(\frac{1}{10}\,m\,\mathrm{Pi}\right)t,\ t=-5..5\right]\right),\ m=-5..5\right):$ # *egyenesek*

> $k := seq(plot([m\cos(t),\ m\sin(t),\ t=0..2\,\mathrm{Pi}]),\ m=0..5):$ # *körök*

```
> plots[display]([e,k],scaling=constrained,title="Koncentrikus
```

```
körök ortogonális trajektóriája");
```



We put the graphs of the circles and lines in the e and k variables then had them displayed in the same coordinate system by the display procedure.

A question arises at this point: are there any other classes of curves with the same properties besides the circles? Similar uniqueness issues and the usage of the free parameters seem to refer to the fact that we are facing with differential equations.

We recommend the following method to solve the task. First, work out a differential equation the general solution of which is the initial class of ellipses. Then change the slope of the tangent to the slope of the perpendicular line in this differential equation and solve the equation generated. The class of curves, which serves as a general solution, provides the orthogonal trajectory of the class of ellipses. So let's see it.

> $x^2 + 3\,y(x)^2 = c^2$ # *görbe_sereg*

$$x^2 + 3\,y(x)^2 = c^2 \quad \textbf{(1)}$$

> $\frac{\mathrm{d}}{\mathrm{d}x}\,\%$

(2)

$$2x + 6y(x)\left(\frac{d}{dx}y(x)\right) = 0 \quad (2)$$

Notice that in the first command we used the y(x) instead of the y which denotes that the y is the function of the x. This technique is generally used when we work with differential equations. We wrote the equation of the ellipses and differentiated it by the x in the second command. As we can see, the system created the derivative of the function correctly. What would have been the solution of the derivation if we had used the y in the first command? In this case, there would be 0 instead of the derivative of the y because the system would have considered the y as the c constant.

The differential equation of the ellipses can be expressed from the equation above with a simple mathematical sorting of equation. We can express the derivative of the y from it.

> $\frac{d}{dx}y(x) = solve\left(\%, \frac{d}{dx}y(x)\right)$

$$\frac{d}{dx}y(x) = -\frac{1}{3}\frac{x}{y(x)} \quad (3)$$

According to this, the slope of the tangent of the ellipse is m=-x/3y at the [x,y] coordinate point and the slope of the perpendicular line is -1/m. Naturally the latter is the slope of the tangent of the perpendicular class of curves at the [x,y] point.

Now let's write the differential equation of the perpendicular class of curves and solve it with the dsolve procedure. Its syntax is easy to remember because its parameters are similar to those of the solve procedure. The first parameter of the dsolve is the differential equation to be solved and its second parameter is the unknown y(x) function.

> $\frac{d}{dx}y(x) = -\frac{1}{rhs(\textbf{(3)})}$

$$\frac{d}{dx}y(x) = \frac{3y(x)}{x} \quad (4)$$

> $dsolve(\textbf{(4)}, y(x))$

$$y(x) = \_C1\, x^3 \quad (5)$$

We received the general solution of the differential equation. The solution is the one-parameter class of third-degree curves. This class of curves is the orthogonal trajectory of the class of ellipses mentioned in the task. We will get back to the illustration of the task later. Now let's discuss the _C1 variable.

Maple is often forced to create new variables during runtime. Previously we became familiar with the _Z variable used in the RootOf notation. In our case it has to denote the constant multiplier appearing in the general solution of the differential equation. Since the system generates variable names that start with the underscore character (_) during runtime we should not start the name of our variables with this character because we can easily get confused. You are highly recommended to avoid this situation.

The graphic display of the two classes of curves is ahead. Probably most of the readers would suggest that we should write procedures for the drawing of the ellipses and power functions. This would be an appropriate way but we know a better one. The DEtools package is the collection of such procedures. Leave the colon from the end of the following instruction to see the procedures of the DEtools.

> $with(DEtools):$

The DEplot procedure displays the solutions of the firs-order differential equation and the tangent field of the solutions in the same coordinate system. Obviously we should use this.

```
> DEplot(diff(y(x), x) = -x/(3*y(x)), y(x), x = -2 .. 2, {[y(1)=1/2], [y(1)=1], [y(1)=3/2], [y(1)=(-1)/2],[y(1)=-1], [y(1)=(-3)/2]}, scaling=constrained, arrows=SMALL, thickness=2, linecolor=blue);
```



Well, although we should see a class of ellipses the graph is a bit crisscrossed. Why is the DEplot drawing such strange graphs? Maybe because the differential equation of (4) is singular. This means that the right side of the differential equation at the point of the x axis is meaningless. In this case the denominator is 0. The tangents of the ellipses at the common intersection with the x axis are parallel with the y axis thus their slopes do not exist in this case. Briefly, the y cannot be expressed as the function of the x at these points.

What is the solution in this case? Let’s approach the issue parametrically. Assume that both the x and y are the functions of the t variable.

> $x(t)^2 + 3\,y(t)^2 = c^2$ # *görbe sereg*

$$x(t)^2 + 3\,y(t)^2 = c^2 \quad (6)$$

We can get the differential equation of the class of curves by the derivation of the equation by the t. Notice that the singularity of the differential equation can be ceased if the derivative of the x is substituted with the y. In this case we can simplify the equation with the y.

> $de_0 := \frac{d}{dt}\ \%$

$$de_0 := 2\,x(t)\left(\frac{d}{dt}\,x(t)\right) + 6\,y(t)\left(\frac{d}{dt}\,y(t)\right) = 0 \quad \textbf{(7)}$$

> $h_1 := \frac{d}{dt}\,x(t) = y(t)$

$$h_1 := \frac{d}{dt}\,x(t) = y(t) \quad \textbf{(8)}$$

> $subs(h_1, de_0)$

$$2\,x(t)\,y(t) + 6\,y(t)\left(\frac{d}{dt}\,y(t)\right) = 0 \quad \textbf{(9)}$$

> $h_2 := \frac{d}{dt}\,y(t) = solve\left(\%, \frac{d}{dt}\,y(t)\right)$

$$h_2 := \frac{d}{dt}\,y(t) = -\frac{1}{3}\,x(t) \quad \textbf{(10)}$$

The h1 and h2 make a system of differential equation concerning the x(t) and y(t) unknown functions the solutions of which are the ellipses appearing in the task.

Use the DEplot procedure of the DEtools package to draw the solutions. But before this we have to give the initial values for which we would like to get the solutions.

> $e := [h_1, h_2]$

$$e := \left[\frac{d}{dt}\,x(t) = y(t),\ \frac{d}{dt}\,y(t) = -\frac{1}{3}\,x(t)\right] \quad \textbf{(11)}$$

> $k := \left\{\left[0, 1, \frac{1}{2}\right], [0, 1, 1], \left[0, 1, \frac{3}{2}\right], [0, 1, 2], \left[0, 1, \frac{5}{2}\right], [0, 1, 3]\right\}$

$$k := \left\{\left[0, 1, \frac{1}{2}\right], [0, 1, 1], \left[0, 1, \frac{3}{2}\right], [0, 1, 2], \left[0, 1, \frac{5}{2}\right], [0, 1, 3]\right\} \quad \textbf{(12)}$$

> $R := DEplot(e, [x(t), y(t)], t = 0..15, k, stepsize = 0.1, scene = [x, y], x = -3..3, y = -3..3, title = "Ellipszisek", scaling = constrained, thickness = 2, linecolor = blue, arrows = none):$

> R



The call sequence of the DEplot procedure is

$$DEplot(diffegyenlet,\ `változók`,\ t\_intervallum,\ kezdeti\ `értékek`,\ \langle `opciók \rangle),$$

or

$$DEplot(diffegyenlet,\ `változók`,\ t\_intervallum,\ x\_intervallum,\ y\_intervallum,\ \langle `opciók \rangle).$$

The differential equation can consist of a one, two or even more dimensional system or a single higher order differential equation. The variables contain the names of the unknown functions. The t_interval is the calculation interval of the independent variable. The initial values have to be determined by the type of the differential equation. In our case the set consists of three-element lists. The first, second and third elements of the list give the initial value of the t, the x and the y. The x and y intervals of the representation are not needed to be given.

We can use the options of the plot. Maple uses the stepsize option for the numeric calculation of the integral curve. This option gives the step size of the independent variable. If it does not appear then the default is [képlet] in which case [a,b] is the interval of the t.

The scene=[x,y] or the scene=[x,y,t] option gives the plot domain of the solution. In the first case the system draws the curve in the x,y plane and t is the parameter. In the second case we get a 3-D curve. This option is useless in the case of a first-order equation. The thickness=n option determines the thickness of the line. The n can be 0,1,2 and 3. The default is 0.

Let’s get to know the procedures of the DEtools package better and draw the direction field of the ellipses with the help of the dfieldplot procedure.

```
> m[1]:=dfieldplot([diff(x(t),t)=y(t), diff(y(t),t)=-1/3*x(t)],[x
  (t),y(t)], t=0..1, x=-3..3, y=-3..3, arrows=SLIM, scaling=
  constrained, title="Iránymezö"):
```

> m_1



The parameters of the dfieldplot and the DEplot procedures are similar. The arrows option is responsible for the display of the arrows of the vector field. The following arrows can be given:

- thin
- slim
- thick
- line
- none

So far we have created two plot objects which we can display with the display procedure in the same coordinate system.

```
> plots[display]({R,m[1]}, title = `Ellipszisek és iránymezejük`);
```



Notice how smoothly the ellipses osculate into the direction field. This last graph could have been created by the DEplot procedure if we had used the arrows option. However, in this case we could not make a difference between the line thickness of the ellipses and the arrows.

Let's continue the illustration of the class of curves and deduce the parametric system of the differential equation of the perpendicular class of curves.

$$> de_1 := subs\left(\left\{\frac{d}{dt}\,x(t) = -\frac{d}{dt}\,y(t),\ \frac{d}{dt}\,y(t) = \frac{d}{dt}\,x(t)\right\}, de_0\right)$$

$$de_1 := -2\,x(t)\left(\frac{d}{dt}\,y(t)\right) + 6\,y(t)\left(\frac{d}{dt}\,x(t)\right) = 0 \tag{13}$$

As you can see we have received the differential equation of the orthogonal trajectory from the differential equation of the ellipse by having exchanged the coordinates of the tangent of the parametric plane curve and given an opposite sign to it. So we have executed the following substitutions:

$$\frac{d}{dt}\,x(t) = -\frac{d}{dt}\,y(t)\,,$$

$$\frac{d}{dt}\,y(t) = \frac{d}{dt}\,x(t)\,.$$

Cease the singularity of the differential equation received.

> $h_1 := \frac{d}{dt}\,y(t) = y(t)$

$$h_1 := \frac{d}{dt}\,y(t) = y(t) \qquad \textbf{(14)}$$

> $subs(h_1, de_1)$

$$-2\,x(t)\,y(t) + 6\,y(t)\left(\frac{d}{dt}\,x(t)\right) = 0 \qquad \textbf{(15)}$$

> $h_2 := isolate\left(\%, \frac{d}{dt}\,x(t)\right)$

$$h_2 := \frac{d}{dt}\,x(t) = \frac{1}{3}\,x(t) \qquad \textbf{(16)}$$

> $ortogonal := [h_2, h_1]$

$$ortogonal := \left[\frac{d}{dt}\,x(t) = \frac{1}{3}\,x(t),\ \frac{d}{dt}\,y(t) = y(t)\right] \qquad \textbf{(17)}$$

We entered the h1 variable then substituted it into the differential equation of the orthogonal trajectory. Then we expressed the derivative of the x(t) from the equation and received the system of differential equation of the orthogonal trajectory provided by the [h2, h1] list.

First, let's draw the vector field with the dfieldplot procedure.

```
> m[2] := dfieldplot(ortogonal, [x(t), y(t)], t = 0 .. 1, x = -3 ..
  3, y = -3 .. 3, arrows = SLIM, scaling = constrained, title =
  `Meröleges vektormezö`):
```

> m_2



We can get an interesting graph if we display the two vector fields drawn with thin arrows in the same coordinate system. Notice that every arrow has its perpendicular counterpart.

> $plots_{display}(\{m_2, m_1\})$



To draw the graphs of the solutions we have to give the initial values from which the solutions originate then display the perpendicular class of curves with the DEplot procedure.

> $k := \left\{\left[0, -1, \frac{1}{8}\right], \left[0, -1, \frac{1}{4}\right], \left[0, -1, \frac{1}{2}\right], \left[0, 1, \frac{1}{2}\right], [0, 1, 1], [0, 1, 2], [0, -1, 1], [0, -1, 2], \left[0, -1, \frac{-1}{8}\right], \left[0, -1, \frac{-1}{4}\right], \left[0, -1, \frac{-1}{2}\right], [0, -1, -1], [0, -1, -2], \left[0, 1, \frac{1}{8}\right], \left[0, 1, \frac{1}{4}\right], \left[0, 1, \frac{-1}{8}\right], \left[0, 1, \frac{-1}{4}\right], \left[0, 1, \frac{-1}{2}\right], [0, 1, -1], [0, 1, -2]\right\}:$

```
> Orto:=DEplot(ortogonal, [x(t),y(t)], -5..5, k, stepsize=.2,
  scene=[x, y], x=-3..3, y=-3..3, title= `Meröleges görbesereg`,
  scaling=constrained, thickness=2, linecolor=blue, arrows=none):
```

> *Orto*



The common display of the solutions and the vector field perfectly illustrates how smoothly the class of curves osculates into its direction field.

```
> DEplot(ortogonal, [x(t),y(t)], -5..5, k, stepsize=.2, scene=[x,
  y], x=-3..3, y=-3..3, title= "Ortogonális trajektóriák és a
  vektormezö", scaling=constrained, thickness=2, linecolor=blue);
```



Let’s dot the i’s and cross the t’s. Display the perpendicular class of curves of the ellipses and orthogonal trajectories in the same coordinate system.

```
> plots[display]({R, Orto}, title=`Ellipszisek és meröleges
  görbeseregük`);
```



>

We have solved the task. We have become familiar with the perpendicular class of curves of the ellipses by having used the DEtools package. .

What Have You Learnt About Maple?

- We can differentiate expressions and equations with the diff procedure. Its syntax is $\frac{d}{dx}$ *egyenlet*, In the case of 1-D Math the input is `diff(egyenlet,x).`
- The dsolve procedure is used for the solution of the differential equations. Its syntax is [képlet]. We get the solution in the form of an equality if there is an explicit or implicit solution to the y(x).
- The DEplot procedure draws solutions that belong to certain initial values. The DEplot can be applied for an arbitrary system of differential equation. Its syntax is
-

$DEplot\big(\text{`differenciálegyenletrendszer`}, \text{`változók`}, t = t_0 .. t_1, \text{kezdeti `értékek`}, x = a .. b, y = c .. d,$

`további` `opciók`).

- The dfieldplot procedure gives the drawing of the direction field of the 1-D or 2-D differential equations. Its syntax is

$$dfieldplot([f(x, y), g(x, y)], [x, y], a..b, x = c..d, y = e..f, arrows = a\ nyilak\ tipusa).$$

Exercises

1. Find the orthogonal trajectories of the following class of curves and draw them in the same coordinate system with the class of curves.

$y = m x$	$x^2 + y^2 = 2 a x$
$y = a x^2;$	$x^2 + y^2 = 2 a y$
$y^2 + 2 a x = 0;$	$x^2 - y^2 = a^2$
$x y = c;$	$y = c e^{-x}$

2. Solve the parametric system of differential equation of the class of ellipses. Prove that the solution satisfies the $x^2 + 3 y^2 = c^2$ equation.

3. Solve the parametric system of differential equation called orthogonal of the orthogonal trajectories. (see (17)) Prove that the solution received satisfies the $y = c x^3$ equation.